<commit_before>{ stdenv, fetchurl, unzip, m4, bison, flex, openssl, zlib }:

stdenv.mkDerivation rec {
  name = "gsoap-${version}";
  version = "2.8.42";

  src = fetchurl {
    url = "mirror://sourceforge/project/gsoap2/gsoap-2.8/gsoap_${version}.zip";
    sha256 = "0fav4lhdibwggkf495pggmqna632jxkh6q2mi32b9hsn883pg5m7";
  };

  buildInputs = [ unzip m4 bison flex openssl zlib ];

  meta = with stdenv.lib; {
    description = "C/C++ toolkit for SOAP web services and XML-based applications";
    homepage = "http://www.cs.fsu.edu/~engelen/soap.html";
    # gsoap is dual/triple licensed (see homepage for details):
    # 1. gSOAP Public License 1.3 (based on Mozilla Public License 1.1).
    #    Components NOT covered by the gSOAP Public License are:
    #     - wsdl2h tool and its source code output,
    #     - soapcpp2 tool and its source code output,
    #     - UDDI code,
    #     - the webserver example code in gsoap/samples/webserver,
    #     - and several example applications in the gsoap/samples directory.
    # 2. GPLv2 covers all of the software
    # 3. Proprietary commercial software development license (removes GPL
    #    restrictions)
    license = licenses.gpl2;
    platforms = platforms.linux;
    maintainers = [ maintainers.bjornfor ];
  };
}<commit_msg>gsoap: 2.8.42 -> 2.8.49<commit_after>{ stdenv, fetchurl, autoreconfHook, unzip, m4, bison, flex, openssl, zlib }:

let
  majorVersion = "2.8";

in stdenv.mkDerivation rec {
  name = "gsoap-${version}";
  version = "${majorVersion}.49";

  src = fetchurl {
    url = "mirror://sourceforge/project/gsoap2/gsoap-${majorVersion}/gsoap_${version}.zip";
    sha256 = "<KEY>";
  };

  buildInputs = [ openssl zlib ];
  nativeBuildInputs = [ autoreconfHook bison flex m4 unzip ];
  # Parallel building doesn't work as of 2.8.49
  enableParallelBuilding = false;

  # Future versions of automake require subdir-objects if the source is structured this way
  # As of 2.8.49 (maybe earlier) this is needed to silence warnings
  prePatch = ''
    substituteInPlace configure.ac \
      --replace 'AM_INIT_AUTOMAKE([foreign])' 'AM_INIT_AUTOMAKE([foreign subdir-objects])'
  '';

  meta = with stdenv.lib; {
    description = "C/C++ toolkit for SOAP web services and XML-based applications";
    homepage = "http://www.cs.fsu.edu/~engelen/soap.html";
    # gsoap is dual/triple licensed (see homepage for details):
    # 1. gSOAP Public License 1.3 (based on Mozilla Public License 1.1).
    #    Components NOT covered by the gSOAP Public License are:
    #     - wsdl2h tool and its source code output,
    #     - soapcpp2 tool and its source code output,
    #     - UDDI code,
    #     - the webserver example code in gsoap/samples/webserver,
    #     - and several example applications in the gsoap/samples directory.
    # 2. GPLv2 covers all of the software
    # 3. Proprietary commercial software development license (removes GPL
    #    restrictions)
    license = licenses.gpl2;
    platforms = platforms.linux;
    maintainers = with maintainers; [ bjornfor ];
  };
}
{'fraction_non_alphanumeric': 0.10875244937949052, 'fraction_numerical': 0.025473546701502287, 'mean_word_length': 3.1058981233243967, 'pattern_counts': {'":': 0, '<': 4, '<?xml version=': 0, '>': 5, 'https://': 0, 'lorem ipsum': 0, 'www.': 2, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '6603947', 'n_tokens_mistral': 1048, 'n_tokens_neox': 994, 'n_words': 297}